Доброго дня.

Подскажите пожалуйста как вывести последнюю цену из определенного периода, никак не соображу. Максимум Период была попытка, не срабатывает.
Если за месяц вывожу - все норм, если за 2 -3 месяца, то он выводит все цены за эти месяцы.
```1c
    ВЫБРАТЬ РАЗРЕШЕННЫЕ
    РеализацияТоваровУслугТовары.Ссылка.Дата КАК Дата,
    РеализацияТоваровУслугТовары.Ссылка КАК Реализация,
    РеализацияТоваровУслугТовары.Номенклатура КАК Номенклатура,
    РеализацияТоваровУслугТовары.Количество КАК Количество,
    РеализацияТоваровУслугТовары.СуммаСНДС КАК СуммаСНДС,
    РеализацияТоваровУслугТовары.Ссылка.Организация КАК Организация
    ПОМЕСТИТЬ втРеализация
    ИЗ
    Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары
    ГДЕ
    РеализацияТоваровУслугТовары.Ссылка.Проведен = &Проведен
    {ГДЕ
    (РеализацияТоваровУслугТовары.Ссылка.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания)}
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    втРеализация.Номенклатура КАК Номенклатура,
    СУММА(втРеализация.Количество) КАК Количество,
    втРеализация.Дата КАК Дата,
    втРеализация.Реализация КАК Реализация,
    ЦеныНоменклатуры.Период КАК Период,
    ЦеныНоменклатуры.Цена КАК Цена,
    втРеализация.СуммаСНДС КАК СуммаСНДС,
    втРеализация.Организация КАК Организация,
    ЦеныНоменклатуры.ВидЦены КАК ВидЦены
    ПОМЕСТИТЬ втЦены
    ИЗ
    втРеализация КАК втРеализация
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры КАК ЦеныНоменклатуры
    ПО втРеализация.Номенклатура = ЦеныНоменклатуры.Номенклатура
    И втРеализация.Дата >= ЦеныНоменклатуры.Период

    СГРУППИРОВАТЬ ПО
    втРеализация.Номенклатура,
    втРеализация.Реализация,
    ЦеныНоменклатуры.Цена,
    втРеализация.СуммаСНДС,
    втРеализация.Организация,
    ЦеныНоменклатуры.ВидЦены,
    ЦеныНоменклатуры.Период,
    втРеализация.Дата
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    втЦены.Номенклатура КАК Номенклатура,
    втЦены.Количество КАК Количество,
    втЦены.Дата КАК Дата,
    втЦены.Период КАК Период,
    втЦены.Цена КАК Цена,
    втЦены.СуммаСНДС КАК СуммаСНДС,
    втЦены.ВидЦены КАК ВидЦены,
    втЦены.Реализация КАК Реализация,
    втЦены.Организация КАК Организация
    ИЗ
    втЦены КАК втЦены

    ГДЕ втЦены.Период >= &ДатаНачала
```
---
Вопрос снят. Сделала. Спасибо кто откликнулся.

```1c
    ВЫБРАТЬ РАЗРЕШЕННЫЕ
    РеализацияТоваровУслугТовары.Ссылка.Дата КАК Дата,
    РеализацияТоваровУслугТовары.Ссылка КАК Реализация,
    РеализацияТоваровУслугТовары.Номенклатура КАК Номенклатура,
    РеализацияТоваровУслугТовары.Количество КАК Количество,
    РеализацияТоваровУслугТовары.СуммаСНДС КАК СуммаСНДС,
    РеализацияТоваровУслугТовары.Ссылка.Организация КАК Организация
    ПОМЕСТИТЬ втРеализация
    ИЗ
    Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары
    ГДЕ
    РеализацияТоваровУслугТовары.Ссылка.Проведен = &Проведен
    {ГДЕ
    (РеализацияТоваровУслугТовары.Ссылка.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания)}
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    МАКСИМУМ(ЦеныНоменклатуры.Период) КАК Период,
    втРеализация.Дата КАК Дата,
    втРеализация.Реализация КАК Реализация,
    втРеализация.Номенклатура КАК Номенклатура,
    втРеализация.Количество КАК Количество,
    втРеализация.СуммаСНДС КАК СуммаСНДС,
    втРеализация.Организация КАК Организация
    ПОМЕСТИТЬ втЦены
    ИЗ
    втРеализация КАК втРеализация
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры КАК ЦеныНоменклатуры
    ПО втРеализация.Номенклатура = ЦеныНоменклатуры.Номенклатура
    И втРеализация.Дата >= ЦеныНоменклатуры.Период

    СГРУППИРОВАТЬ ПО
    втРеализация.Дата,
    втРеализация.Реализация,
    втРеализация.Номенклатура,
    втРеализация.Количество,
    втРеализация.СуммаСНДС,
    втРеализация.Организация
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    втЦены.Дата КАК Дата,
    втЦены.Номенклатура КАК Номенклатура,
    втЦены.Организация КАК Организация,
    втЦены.Количество КАК Количество,
    втЦены.СуммаСНДС КАК СуммаСНДС,
    втЦены.Реализация КАК Реализация,
    ЦеныНоменклатуры.Цена КАК Цена,
    ЦеныНоменклатуры.ВидЦены КАК ВидЦены,
    ЦеныНоменклатуры.Период КАК Период
    ИЗ
    втЦены КАК втЦены
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры КАК ЦеныНоменклатуры
    ПО втЦены.Номенклатура = ЦеныНоменклатуры.Номенклатура
    И втЦены.Период = ЦеныНоменклатуры.Период
```
---
вывести последнюю цену из определенного периода
Используйте виртуальную таблицу регистра сведений СрезПоследних на дату окончания.

```1c
    ВЫБРАТЬ
    втРеализация.Дата КАК Дата,
    втРеализация.Реализация КАК Реализация,
    втРеализация.Номенклатура КАК Номенклатура,
    втРеализация.Количество КАК Количество,
    втРеализация.СуммаСНДС КАК СуммаСНДС,
    втРеализация.Организация КАК Организация,
    ЦеныНоменклатурыСрезПоследних.Цена КАК Цена
    ИЗ
    втРеализация КАК втРеализация
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры.СрезПоследних(&ДатаОкончания, ) КАК ЦеныНоменклатурыСрезПоследних
    ПО (втРеализация.Номенклатура = ЦеныНоменклатурыСрезПоследних.Номенклатура)
```
---
Это если последнюю цену, если как в названии темы на дату документа, то есть множество примеров реализации, нужно в поиске найти "Цена на дату документа".
---
https://infostart.ru/1c/reports/1457834/
---
Вопрос снят. Сделала. Спасибо кто откликнулся.

```1c
    ВЫБРАТЬ РАЗРЕШЕННЫЕ
    РеализацияТоваровУслугТовары.Ссылка.Дата КАК Дата,
    РеализацияТоваровУслугТовары.Ссылка КАК Реализация,
    РеализацияТоваровУслугТовары.Номенклатура КАК Номенклатура,
    РеализацияТоваровУслугТовары.Количество КАК Количество,
    РеализацияТоваровУслугТовары.СуммаСНДС КАК СуммаСНДС,
    РеализацияТоваровУслугТовары.Ссылка.Организация КАК Организация
    ПОМЕСТИТЬ втРеализация
    ИЗ
    Документ.РеализацияТоваровУслуг.Товары КАК РеализацияТоваровУслугТовары
    ГДЕ
    РеализацияТоваровУслугТовары.Ссылка.Проведен = &Проведен
    {ГДЕ
    (РеализацияТоваровУслугТовары.Ссылка.Дата МЕЖДУ &ДатаНачала И &ДатаОкончания)}
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    МАКСИМУМ(ЦеныНоменклатуры.Период) КАК Период,
    втРеализация.Дата КАК Дата,
    втРеализация.Реализация КАК Реализация,
    втРеализация.Номенклатура КАК Номенклатура,
    втРеализация.Количество КАК Количество,
    втРеализация.СуммаСНДС КАК СуммаСНДС,
    втРеализация.Организация КАК Организация
    ПОМЕСТИТЬ втЦены
    ИЗ
    втРеализация КАК втРеализация
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры КАК ЦеныНоменклатуры
    ПО втРеализация.Номенклатура = ЦеныНоменклатуры.Номенклатура
    И втРеализация.Дата >= ЦеныНоменклатуры.Период

    СГРУППИРОВАТЬ ПО
    втРеализация.Дата,
    втРеализация.Реализация,
    втРеализация.Номенклатура,
    втРеализация.Количество,
    втРеализация.СуммаСНДС,
    втРеализация.Организация
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    втЦены.Дата КАК Дата,
    втЦены.Номенклатура КАК Номенклатура,
    втЦены.Организация КАК Организация,
    втЦены.Количество КАК Количество,
    втЦены.СуммаСНДС КАК СуммаСНДС,
    втЦены.Реализация КАК Реализация,
    ЦеныНоменклатуры.Цена КАК Цена,
    ЦеныНоменклатуры.ВидЦены КАК ВидЦены,
    ЦеныНоменклатуры.Период КАК Период
    ИЗ
    втЦены КАК втЦены
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрСведений.ЦеныНоменклатуры КАК ЦеныНоменклатуры
    ПО втЦены.Номенклатура = ЦеныНоменклатуры.Номенклатура
    И втЦены.Период = ЦеныНоменклатуры.Период
```